

Differentiaaliyhtälöryhmän numeerinen ratkaiseminen

Ensimmäisen kertaluvun differentiaaliyhtälöryhmä — vaikkapa korkeamman kertaluvun yhtälöä vastaava normaaliryhmä — voidaan ratkaista numeerisesti täsmälleen samanlaisilla kaavoilla kuin ensimmäisen kertaluvun yhtälö $y' = f(x, y)$. Edellytyksenä luonnollisesti on, että alkuehto on annettu.

Seuraavat koodit ovat täsmälleen samat kuin ensimmäisen kertaluvun yhtälöä koskevassa esimerkissä. (Solut on ajettava, jotta määritelmät tulevat voimaan.)

```
euler[f_, x0_, y0_, h_, xend_] := Module[{X, Y, kend},
  X[0] = x0;
  X[k_] := X[k] = X[k - 1] + h;
  Y[0] = y0;
  Y[k_] := Y[k] = Y[k - 1] + h f[X[k - 1], Y[k - 1]];
  kend = Round[(xend - x0) / h];
  Table[{X[k], Y[k]}, {k, 0, kend}]]
```

General::spell1 :

Possible spelling error: new symbol name "kend" is similar to existing symbol "xend". MORE...

```
pareuler[f_, x0_, y0_, h_, xend_] := Module[{X, Y, kend},
  X[0] = x0;
  X[k_] := X[k] = X[k - 1] + h;
  Y[0] = y0;
  Y[k_] :=
    Y[k] = Y[k - 1] + h / 2 (f[X[k - 1], Y[k - 1]] + f[X[k], Y[k - 1] + h f[X[k - 1], Y[k - 1]]]);
  kend = Round[(xend - x0) / h];
  Table[{X[k], Y[k]}, {k, 0, kend}]]
```

```
rungekutta[f_, x0_, y0_, h_, xend_] := Module[{X, Y, kend},
  X[0] = x0;
  X[k_] := X[k] = X[k - 1] + h;
  Y[0] = y0;
  Y[k_] := Y[k] = Y[k - 1] + 1 / 6 Module[{k1, k2, k3, k4}, k1 = h f[X[k - 1], Y[k - 1]];
    k2 = h f[X[k - 1] + h / 2, Y[k - 1] + k1 / 2];
    k3 = h f[X[k - 1] + h / 2, Y[k - 1] + k2 / 2];
    k4 = h f[X[k - 1] + h, Y[k - 1] + k3];
    k1 + 2 k2 + 2 k3 + k4];
  kend = Round[(xend - x0) / h];
  Table[{X[k], Y[k]}, {k, 0, kend}]]
```

```

adamsbashforth[f_, x0_, y0_, h_, xend_] := Module[{X, Y, kend},
  X[0] = x0;
  X[k_] := X[k] = X[k - 1] + h;
  Y[0] = y0;
  Y[k_ /; k <= 3] :=
    Y[k] = Y[k - 1] + 1 / 6 Module[{k1, k2, k3, k4}, k1 = h f[X[k - 1], Y[k - 1]];
      k2 = h f[X[k - 1] + h / 2, Y[k - 1] + k1 / 2];
      k3 = h f[X[k - 1] + h / 2, Y[k - 1] + k2 / 2];
      k4 = h f[X[k - 1] + h, Y[k - 1] + k3];
      k1 + 2 k2 + 2 k3 + k4];
  Y[k_ /; k > 3] := Y[k] = Y[k - 1] + h / 24 (55 f[X[k - 1], Y[k - 1]] -
    59 f[X[k - 2], Y[k - 2]] + 37 f[X[k - 3], Y[k - 3]] - 9 f[X[k - 4], Y[k - 4]]);
  kend = Round[(xend - x0) / h];
  Table[{X[k], Y[k]}, {k, 0, kend}]

```

Esimerkkinä olkoon differentiaaliyhtälö $y'' + xy = 0$ alkuehtona $y(0) = 0$, $y'(0) = 1$. Kyseessä on Airyn yhtälö, jossa x-akselin suunta on käännetty, ts. x :n merkki on vaihdettu. Yhtälöä vastaava normaaliryhmä muodostuu kahdesta yhtälöstä $y' = z$, $z' = -xy$ alkuehtona $y(0) = 0$, $z(0) = 1$. Vektorimuodossa normaaliryhmä on $Y' = F(x, Y)$; oikean puolen vektoriarvoinen funktio määritellään Mathematicalle seuraavasti:

```
f[x_, {y_, z_}] := {z, -x y}
```

Askelpituus ja arvo, jolla alkuehto annetaan:

```

h = 0.1
0.1
x0 = 0
0

```

Tuntemattomana funktiona on vektori $Y = (y, z)$, joten alkuarvo on myös vektori:

```

y0 = {0, 1}
{0, 1}

```

Lasketaan välillä $[0, 10]$:

```

xend = 10
10

```

Ratkaisufunktion approksimaatio jokaisessa pisteessä x_k saadaan kaksikomponenttisena vektorina. Edellinen komponentti on itse funktion y arvo, jälkimmäinen sen derivaatan $z = y'$ arvo:

```
eulerdata = euler[f, x0, y0, h, xend];
Short[eulerdata, 10]

{{0, {0, 1}}, {0.1, {0.1, 1}}, {0.2, {0.2, 0.999}}, {0.3, {0.2999, 0.995}},
 {0.4, {0.3994, 0.986003}}, {0.5, {0.498, 0.970027}}, {0.6, {0.595003, 0.945127}},
 <<88>>, {9.5, {2.79772, 17.5359}}, {9.6, {4.55131, 14.8781}}, {9.7, {6.03912, 10.5088}},
 {9.8, {7.09, 4.65089}}, {9.9, {7.55509, -2.29731}}, {10., {7.32536, -9.77684}}
```

Muuttujaan `eulerdata` talletettu tulostus on huomattavan pitkä, ja siitä on Mathematican `Short`-funktion avulla kirjoitettu näkyviin vain osa. Merkintä `<<. . . >>` osoittaa, kuinka monta termiä on jätetty pois.

Tästä voidaan poimia vain argumentin x ja funktion y arvot sijoitussäännöllä, minkä jälkeen voidaan piirtää kuva:


```
poiminta = {x_, {y_, z_}} :> {x, y}

{x_, {y_, z_}} => {x, y}

eulerdata = eulerdata /. poiminta;
Short[eulerdata, 10]

{{0, 0}, {0.1, 0.1}, {0.2, 0.2}, {0.3, 0.2999}, {0.4, 0.3994},
 {0.5, 0.498}, {0.6, 0.595003}, {0.7, 0.689516}, {0.8, 0.780458}, <<83>>,
 {9.2, -2.36349}, {9.3, -0.813264}, {9.4, 0.954409}, {9.5, 2.79772},
 {9.6, 4.55131}, {9.7, 6.03912}, {9.8, 7.09}, {9.9, 7.55509}, {10., 7.32536}}
```

```
eulerkuva = ListPlot[eulerdata]
```




- Graphics -

Vastaavat laskut muita menetelmiä käyttäen:

```
pareulerdata = pareuler[f, x0, y0, h, xend] /. poiminta;
Short[pareulerdata, 10]

{{0, 0}, {0.1, 0.1}, {0.2, 0.1999}, {0.3, 0.2994}, {0.4, 0.398002},
 {0.5, 0.495011}, {0.6, 0.589542}, {0.7, 0.680533}, {0.8, 0.766754},
 <<84>>, {9.3, 0.349574}, {9.4, 0.533061}, {9.5, 0.665676}, {9.6, 0.733862},
 {9.7, 0.73008}, {9.8, 0.653771}, {9.9, 0.511658}, {10., 0.317305}}
```

```
pareulerkuva = ListPlot[pareulerdata]
```




- Graphics -

```
rungekuttadata = rungekutta[f, x0, y0, h, xend] /. poiminta;
Short[rungekuttadata, 10]
```

```
{{0, 0}, {0.1, 0.09999917}, {0.2, 0.199867}, {0.3, 0.299325}, {0.4, 0.39787},
 {0.5, 0.494807}, {0.6, 0.589255}, {0.7, 0.680154}, {0.8, 0.76628},
 <<84>>, {9.3, 0.22355}, {9.4, 0.41828}, {9.5, 0.573972}, {9.6, 0.675552},
 {9.7, 0.712792}, {9.8, 0.681459}, {9.9, 0.583908}, {10., 0.429057}}
```

```
rungekuttakuva = ListPlot[rungekuttadata]
```




- Graphics -

```
adamsbashforthdata = adamsbashforth[f, x0, y0, h, xend] /. poiminta;
Short[adamsbashforthdata, 10]
```

```
{{0, 0}, {0.1, 0.09999917}, {0.2, 0.199867}, {0.3, 0.299325}, {0.4, 0.397869},
 {0.5, 0.494803}, {0.6, 0.589247}, {0.7, 0.680138}, {0.8, 0.766252},
 <<84>>, {9.3, 0.208299}, {9.4, 0.40112}, {9.5, 0.556475}, {9.6, 0.659453},
 {9.7, 0.699829}, {9.8, 0.673184}, {9.9, 0.581509}, {10., 0.433199}}
```

```
adamsbashforthkuva = ListPlot[adamsbashforthdata]
```



- Graphics -


Esimerkkinä oleva alkuarvoprobleema voidaan ratkaista myös tarkasti Airyn funktioiden avulla. Mathematica (versio 4.1) ei kuitenkaan osaa ratkaista kyseessä olevaa normaaliryhmää, joten on tarkasteltava alkuperäistä Airyn yhtälöä:

```
tarkkaratkaisu = y /. First[DSolve[{y''[x] + x y[x] == 0, y[0] == 0, y'[0] == 1}, y, x]]
```

```
Function[{x},
```

$$\frac{1}{6} \left(3 (-1)^{2/3} 3^{1/3} \text{AiryAi} [(-1)^{1/3} x] \text{Gamma} \left[\frac{1}{3} \right] - (-1)^{2/3} 3^{5/6} \text{AiryBi} [(-1)^{1/3} x] \text{Gamma} \left[\frac{1}{3} \right] \right)$$


```
tarkkakuva = Plot[tarkkaratkaisu[x], {x, 0, xend}]
```



- Graphics -

Eri menetelmillä saadut approksimaatiot ja tarkka ratkaisu verrattuina:

```
kuva = Show[eulerkuva, pareulerkuva,
  rungekuttakuva, adamsbashforthkuva, tarkkakuva, PlotRange -> All]
```



- Graphics -

Tarkan ratkaisun tarkkuus riippuu luonnollisesti siitä, miten Mathematica laskee ratkaisussa esiintyvien Airyn funktioiden ja gammafunktion arvot.

Vertailun vuoksi alkuarvoprobleema voidaan ratkaista myös Mathematican käyttämällä numeerisen ratkaisemisen algoritmilla, joka on edellä esitettyjä kehittyneempi:

```
numeerinenratkaisu =
  y /. First[NDSolve[{y'[x] + x y[x] == 0, y[0] == 0, y'[0] == 1}, y, {x, x0, xend}]]
InterpolatingFunction[{{0., 10.}}, <>]
```

Tulos on interpolaatiofunktio, jonka arvo välin päätepisteessä voidaan laskea tavalliseen tapaan. Eri ratkaisujen antamat arvot loppupisteessä $x = 10$:

```
{Last[eulerdata][[2]], Last[pareulerdata][[2]], Last[rungekuttadata][[2]],
  Last[adamsbashforthdata][[2]], tarkkaratkaisu[xend] // N, numeerinenratkaisu[xend]}
{7.32536, 0.317305, 0.429057, 0.433199, 0.428719 + 2.10942 × 10-15 i, 0.428719}
```

Linkit

[korkeamman kertaluvun yhtälön numeerinen ratkaiseminen](#)

[Eulerin menetelmä](#)

[parannettu Eulerin menetelmä](#)

[Rungen–Kuttan menetelmä](#)

[Adamsin–Bashforthin menetelmä](#)

[normaaliryhmä](#)

SKK 30.04.2001